

Premiers programmes en OCaml

Guillaume Rousseau
MP2I Lycée Pierre de Fermat
guillaume.rousseau@ens-lyon.fr

14 janvier 2026

Introduction

OCaml est un langage de programmation développé dans les années 1980-1990, en France, par des chercheurs et chercheuses de l'INRIA¹. C'est un langage fonctionnel, présentant des liens forts avec la logique et la théorie de la déduction, ce qui en fait un langage couramment utilisé dans ces domaines de recherche. Il a par exemple été utilisé pour écrire le logiciel **Rocq**, un assistant de preuve permettant de fournir des démonstrations formelles de théorèmes, et de montrer la correction de programmes. Rocq et OCaml ont été utilisés pour écrire et prouver le compilateur C **CompCert** : ce compilateur génère des programmes à peine plus lents que GCC, mais est garanti sans bug².

Le langage donne beaucoup de garanties de sûreté, notamment comparé au C, mais reste relativement performant, il est donc parfois utilisé en industrie pour les systèmes critiques (aéronautique, nucléaire, ...). Facebook utilise également OCaml en interne, et a même développé son propre langage, Reason, mélangeant les notions clés d'OCaml avec des éléments de syntaxe de javascript.

Langage impératif, langage fonctionnel Le C est un langage *impératif*. Cela signifie qu'un programme C est constitué d'instructions, qui disent à la machine ce qu'elle doit faire. En programmation impérative, on parle d'*état du programme* (la mémoire, les variables), et les instructions servent à modifier l'état du programme.

OCaml est un langage *fonctionnel*. La programmation fonctionnelle est un autre paradigme de programmation, dans laquelle un programme est une immense *expression* mathématique que l'on évalue. Dans une telle expression, on peut faire intervenir des fonctions, qui sont des valeurs au même titre que les entiers, les flottants ou les chaînes de caractères.

Dans un langage fonctionnel, il n'y a donc ni variable, ni boucle. Notons qu'OCaml est un langage fonctionnel dit **impur**, car il permet tout de même d'utiliser des éléments de programmation impérative, mais certains langages, comme le Haskell, sont purement fonctionnels !

1. Institut national de recherche en informatique et en automatique

2. Si l'on suppose que Rocq est lui-même correct !

1 Premiers programmes

Un programme OCaml peut être compilé comme du C, ou interprété comme du Python.

ocamlc est le compilateur usuel utilisé pour OCaml. Les fichiers OCaml auront pour extension “.ml”. Pour compiler un programme “mon_programme.ml”, on tape :

```
ocamlc mon_programme.ml -o nom_executable
```

Comme avec gcc, si l’on ne précise pas de nom d’exécutable, celui-ci s’appellera **a.out**.

Pour lancer l’interpréteur OCaml, on peut utiliser la commande **ocaml**, mais on préférera utiliser **utop** qui est un interpréteur plus pratique, avec un historique, de l’auto-complétion, etc... Pour indiquer la fin d’une expression dans l’interpréteur, on utilise `;;`.

Remarque 1

Dans utop, à tout moment, on peut taper CTRL+C pour arrêter la ligne actuelle (en cas de boucle infinie par exemple). On peut aussi taper CTRL+D pour quitter utop.

Q1. Lancer utop

Q2. Taper les expressions suivantes (sans oublier `;;` à chaque fois) :

```
— 5
— 5.67
— true
— 'h'
— "bonjour"
— ()
```

On remarque que lorsqu’on tape une valeur, OCaml détecte tout seul le type, et nous l’affiche. En OCaml, les types de base sont `int`, `float`, `bool`, `char`, `string` et `unit` :

Comparons avec le C :

- `bool` n’est pas compatible avec les entiers, c’est un type à part
- `string` est un type totalement distinct de `char`.
- `unit` est l’équivalent du type `void`. Il possède un unique élément, noté `()` (et aussi prononcé “unit”).

A Opérateurs

Comme tous les langages de programmation, OCaml possède des opérateurs sur les types de base : additions, soustractions, tests d’égalité, d’inégalité, opérations booléennes, etc...

Q3. Taper les expressions suivantes :

```
— 3+8
— 5 = 2+3
— 9.36 < 1.2
— not true
— true && false
— true || false
```

— `"bonjour " ^ "tout le monde"`

Q4. Taper les expressions suivantes :

— `3.65 + 9`

— `3.65 + 2.35`

Que constate-t'on ?

En OCaml, le typage est strict. En particulier, les `int` et les `float` ne peuvent pas être mélangés. Pour additionner des flottants, on doit utiliser l'opérateur `+.` et idem pour la multiplication, la division, la soustraction.


Q5. Taper les expressions suivantes :

— `2.0 +. 0.5`

— `5. /. 2.`

B Arbre de syntaxe

Un programme OCaml est une expression mathématique, que l'on peut représenter sous une forme graphique appelée "arbre de syntaxe". Voici quelques exemples d'expressions et leurs arbres de syntaxe :



On peut donc définir formellement une expression OCaml comme suit :

Définition 1

Une expression OCaml est :

- Une constante d'un type de base (`1`, `2`, `false`, `true`, `3.48`, `0.23`, ...)
- Un opérateur binaire appliqué sur deux expressions OCaml de types compatibles (`(1 + 2)`, `(5-3)+ 4`, `(1 = 2) || false`, ...)
- Un opérateur unaire appliqué sur une expression OCaml de type compatible (`(-12)`, `-(2.32 +. 0.5)`, `not (2 < 5)`)

Cette définition est partielle, on viendra la compléter au fur et à mesure que l'on rencontre de nouveaux éléments de syntaxe d'OCaml.

Une première remarque : cette définition est "récursive" car on définit ce qu'est une expression en fonction des expressions. La terminologie exacte est que c'est une définition **inductive** (Cf prochain chapitre).

Notons aussi qu'avec cette définition, il est ambigu de savoir si `1 + 2 + 3` veut dire `(1+2)+3` ou bien `1+(2+3)` : il y a deux arbres de syntaxe possibles. Par associativité de l'addition, on peut choisir arbitrairement l'un ou l'autre. Si l'on considère l'expression `1 + 2 * 3`, il n'y a pas d'ambiguïté grâce à la précedence des opérateurs : la multiplication est prioritaire sur l'addition, l'expression doit donc être comprise comme `1 + (2 * 3)`.

C Fonctions

Revenons au mélange d'entiers et de flottants. Pour passer de l'un à l'autre, il faut utiliser des fonctions : `float_of_int` et `int_of_float`. En OCaml, **L'application de fonction se fait en écrivant la fonction puis l'argument, séparés d'un espace, sans besoin de parenthèses.**

Q6. Taper :

```
— float_of_int 3
— int_of_float 6.3
— float_of_int 3 +. 6.21
```

On remarque que la dernière expression est comprise comme `(float_of_int 3) +. 6.21` et pas `float_of_int (3 +. 6.21)`. On dit que l'application de fonction est **prioritaire** sur l'addition. L'application de fonction est en réalité prioritaire sur (presque) tout.

En OCaml, **les parenthèses ne servent pas à faire des appels de fonction!** Les parenthèses servent uniquement à **regrouper une expression** pour la rendre "d'un seul tenant". Il est donc inutile d'écrire `float_of_int(5)` car 5 est déjà d'un seul tenant, on peut donc écrire `float_of_int 5` sans parenthèses. En revanche, lorsque l'on écrit `float_of_int (5 + 2)`, les parenthèses sont nécessaires.

Définition 2

On dit qu'une expression est **atomique** si c'est :

- une constante (1, 2, 3.24, true, ...)
- un identifiant (`x`, `y`, ...)
- une expression entre parenthèses (`(2 + 3)`, ...)

OCaml étant un langage fonctionnel, les **fonctions** sont des valeurs comme les autres :

Q7. Taper les expressions suivantes :

```
— int_of_float
— float_of_int
```

Pour la dernière expression, l'interpréteur affiche :

```
- : int -> float = <fun>
```

Dans un type, la flèche `->` se lit "flèche" ou "donne". Une fonction de type `A -> B` prend en entrée un élément de type A et calcule un élément de type B. Par exemple, `float_of_int` prend un int et donne un float.

On peut donc étendre notre définition des expressions OCaml :

Définition 3

Sont également des expressions OCaml :

- Une expression OCaml entre parenthèses
- Deux expressions OCaml atomiques mises l'une après l'autre.

Le dernier cas correspond à une **application de fonction**.

Dans les arbres de syntaxe, on pourra par exemple représenter les applications de fonction par le mot **App**. On pourra traiter une application comme une sorte d'opérateur binaire.

Exemple 1

Voici l'arbre de syntaxe de l'expression `float_of_int (3 + int_of_float (2.5 *. 2.1))`



En OCaml, on peut créer ses propres fonctions. Par exemple, la fonction $f : x \mapsto x + 1$:

1 `(fun x -> x + 1)`

Pour appliquer cette fonction à une autre expression, par exemple $(5 + 3)$, on écrira donc :

1 `(fun x -> x + 1) (5 + 3)`

ce qui affiche bien 9. OCaml vérifie toujours que le type de l'argument d'une fonction correspond bien au type attendu. Par exemple, l'expression suivante générera une erreur :

1 `(fun x -> x + 1) 0.5`

On peut donc mettre à jour notre définition des expressions :


Définition 4

Si A est une expression, alors `fun x -> A` est également une expression. Une telle expression s'appelle une ***abstraction***.

Dans les arbres de syntaxe, on pourra représenter les abstraction par le mot **Fun**.

Exemple 2

Voici l'arbre de syntaxe de l'expression `(fun x -> x + 1)(5 + 3)`



D Évaluation des expressions, environnements

Dans un langage impératif, on dit que l'on exécute une suite d'instructions. Dans langage fonctionnel comme OCaml, on dit que l'on **évalue** une expression. Évaluer une expression, c'est la transformer en une valeur. Par exemple, lorsque l'on transforme $\boxed{3 + 2}$ en 5, on a évalué l'expression

Considérons l'expression $\boxed{3 + 2}$. Cette expression est de type "un opérateur binaire appliqué sur deux expressions". On commence donc par évaluer les deux sous-expressions : $\boxed{3}$ et $\boxed{2}$. Ces deux expressions sont des constantes, elles s'évaluent donc en elles-mêmes. On applique alors l'opérateur $+$ sur les deux valeurs obtenus, on obtient 5.

Considérons maintenant l'expression $\boxed{(3 + 2) * (1 - 4)}$. Cette expression est de type "un opérateur binaire appliqué sur deux expressions". On commence donc par évaluer les deux sous-expressions : $\boxed{(3 + 2)}$ et $\boxed{(1 - 4)}$. Pour évaluer la première, on fait comme précédemment, on obtient alors 5, et de manière analogue, la deuxième expression s'évalue en -3 . Alors, l'expression initiale s'évalue en -15 .

On peut donc donner un algorithme simple permettant d'évaluer une expression OCaml :

Algorithme 1 : Évaluation

Entrée(s) : Un arbre de syntaxe OCaml, c'est à dire une expression E

Sortie(s) : La valeur obtenue en évaluant E

```

1 si  $E$  est une constante  $c$  alors
2   retourner  $c$ 
3 si  $E$  est de la forme  $E_1 + E_2$  alors
4    $v_1 \leftarrow$  Évaluer  $E_1$ ;
5    $v_2 \leftarrow$  Évaluer  $E_2$ ;
6   retourner  $v_1 + v_2$ 
7 si  $E$  est de la forme  $E_1 - E_2$  alors
8    $v_1 \leftarrow$  Évaluer  $E_1$ ;
9    $v_2 \leftarrow$  Évaluer  $E_2$ ;
10  retourner  $v_1 - v_2$ 
// Ainsi de suite
```

Intéressons nous maintenant à l'expression $\boxed{(\text{fun } x \rightarrow x+1)(5+3)}$. Étudions la manière dont OCaml évalue cette expression. Pour commencer, on évalue $5+3$ ce qui donne 8. Ensuite, on doit appliquer la fonction à l'argument 8. Pour cela, on stocke quelque part l'information " \boxed{x} vaut 8", et on évalue le corps de la fonction : $\boxed{x + 1}$. On doit donc évaluer \boxed{x} , et avec l'information que l'on a stocké, on sait que le résultat est 8. Ainsi, on évalue $x + 1$ à 9, et c'est le résultat de l'expression initiale.

Ainsi, lorsque l'on évalue une expression OCaml, on doit stocker des informations sur les différentes variables. L'objet servant à stocker ces informations s'appelle un **environnement** ou un **contexte**.

Exercice 1

Dessiner l'arbre de syntaxe de l'expression $\boxed{2 * 3 + (2 - 1 / 3)}$, et lui appliquer l'algorithme d'évaluation.

Définition 5

Un *environnement* est un ensemble d'associations de la forme $x \mapsto v$, où x est un identifiant (i.e. un nom de variable) et v une valeur.

Étant donné E une expression et ρ un environnement, on peut **évaluer** E dans ρ . Modifions notre algorithme d'évaluation pour prendre en compte l'environnement :

Algorithme 2 : Evaluer(E, ρ)

Entrée(s) : E un arbre de syntaxe, ρ un environnement

Sortie(s) : La valeur obtenue en évaluant E dans ρ

```

1 si  $E$  est une constante  $c$  alors
2   retourner  $c$ 
3 si  $E$  est une variable  $v$  alors
4   retourner la valeur associée à  $v$  dans  $\rho$ 
5 si  $E$  est de la forme  $(\text{fun } x \rightarrow E_1) E_2$  alors
6    $v_2 \leftarrow$  évaluer  $E_2$  dans  $\rho$ ;
7    $\rho' \leftarrow \rho$  mis à jour avec  $x \mapsto v_2$ ;
8   retourner Evaluer( $E_1, \rho'$ )
9 si  $E$  est de la forme  $E_1 + E_2$  alors
10   $v_1 \leftarrow$  Evaluer( $E_1, \rho$ );
11   $v_2 \leftarrow$  Evaluer( $E_2, \rho$ );
12  retourner  $v_1 + v_2$ 

// Ainsi de suite pour les autres opérateurs binaires...
```

Exercice 2

Pour chacune des expressions suivantes, dessiner l'arbre de syntaxe, et appliquer l'algorithme d'évaluation à partir d'un contexte vide.

```

1 (fun a -> a +. 3.) 6.;
2 (fun x -> x + (fun y -> x*y)(x+1)) ((fun x->6-x) 3);;
```

E Fonctions d'ordre supérieur

Considérons la fonction suivante :

```
1 fun f -> 2 * f 2
```

Si l'on tape l'expression précédente dans utop, l'interpréteur nous dit que le résultat est de type `(int -> int)-> int`. Cette fonction prend donc en paramètre une fonction de type `int -> int`, et renvoie un entier. Par exemple :

```
1 (fun f -> 2 * f 2) (fun x-> x + 1) ;;
```

Q8. Dessiner l'arbre de l'expression précédente et appliquer l'algorithme d'évaluation.

Q9. En langage naturel, que fait la fonction `fun f -> 2 * f 2`?

Définition 6

On dit qu'une fonction est *d'ordre supérieur* si elle prend en argument une fonction ou renvoie une fonction.

Exercice 3

Décrire les fonctions suivantes en français.

```
1 fun f -> (fun x -> f (f x)) ;;
2 fun f -> (fun x -> (f (x +. 0.00001) -. f x) /. 0.00001) ;;
3 fun k -> (fun y -> k*y) ;;
```

F Syntaxe “let in”

Il va vite être compliqué de construire des expressions lisibles. Afin de rendre le code plus digeste, on utilise en OCaml une syntaxe qui **ressemble** aux affectations de variable des programmes impératif. Cette syntaxe est construite avec les deux mots-clés `let` et `in`. Par exemple :

```
1 let x = 2 in
2 x + 3;;
```

Pour évaluer une expression de la forme `let x = e1 in e2`, on évalue `e1` en une valeur v_1 puis on évalue `e2` dans l'environnement $(\{x\} \mapsto v_1)$. La preuve que ce n'est pas une affectation de variable est que si l'on essaie d'évaluer x après avoir tapé cette expression, on obtient une erreur car x n'est pas connu. En réalité, x n'existe dans le contexte que lors de l'évaluation de la partie droite du *let in*.

Remarque 2

`let x = e1 in e2` est un raccourci pour dire `(fun x -> e2)e1;;`. Lorsqu'une syntaxe sert simplement à raccourcir le code, ou à faciliter son utilisation, sans rajouter de fonctionnalité, on dit que c'est du *sucre syntaxique*.

On peut imbriquer les *let in* comme on veut :

Q10. Taper les expressions suivantes :

```
1 let x = 3 in
2 let y = 5 in
3 x + y;;
4
5 let x =
6   let t = 3.2 in
7   t *. 5.0
8 in (x < 31.0);;
```

Lorsque l'on écrit `let x = A in B`, on peut voir ça comme rajouter au contexte une valeur pour x , avant de calculer B . Quand on met plusieurs `let in` à la suite, par exemple

```
1 let x = 3 in
2 let y = 5 in
3 x + y;;
```

on construit le contexte en y ajoutant petit à petit des informations sur les variables.

Dans les programmes OCaml et dans l'interpréteur, il existe un contexte global. Pour rajouter une valeur au contexte global dans l'interpréteur, il suffit de remplacer le “in” par “;;”.

Q11. Taper le code suivant dans utop :

```
1 let x = 3;;
2 let y = 5;;
3 x + y ;;
4 let f = fun z -> (x + z);;
5 f 1;;
6 let x = 50;;
7 f 1;;
```

Q12. Que remarque-t'on avec les 4 dernières lignes ?

Remarque 3

let signifie “soit” en anglais, donc `let x = ... ;;` se dirait en français “soit x égal à ...”. Cette syntaxe est donc inspirée du vocabulaire mathématique.

Pour les fonctions, il existe un deuxième niveau de sucre syntaxique, qui consiste à éliminer le mot clé `fun` et la flèche.

Q13. Taper le code suivant :

```
1 let f = fun x -> x * x ;;
2 let g x = x * x;;
```

Regardons la fonction suivante :

```
1 let add x = fun y -> x + y;;
```

cette fonction est de type `int -> (int -> int)`. Autrement dit, elle prend en entrée un entier, et renvoie une fonction. Pour k entier, `add k` est la fonction qui ajoute k à son argument. Donc, on peut écrire :

```
1 let f = add 3;;
2 let x = f 5;;
```

ou même :

```
1 let x = (add 3) 5;;
```

En OCaml, le parenthésage se fait automatiquement à gauche. Autrement dit, les deux expressions suivantes sont équivalentes :

```
1 (add 3) 5;;
2 add 3 5;;
```

Reprenons la fonction `add`. On peut utiliser un nouveau niveau de sucre syntaxique et la définir comme suit :

```
1 let add x y = x + y;;
```

Formellement, cela veut dire que la fonction `add` prend un argument `x`, et renvoie une fonction qui prend elle même un argument `y`, et qui renvoie $x + y$. Informellement, on dira que `add` prend deux arguments, x et y . Cependant, c'est un **abus** de langage !

G Type d'une fonction

Pour une fonction f définie par `let f x1 x2 ... xn = e`, OCaml donne une *signature* à f , qui correspond aux types des paramètres et de la valeur calculée. Par exemple :

```
1 let ma_fonction f x y = f (x + y) + f(x - y);;
```

Cette ligne affiche :

```
val ma_fonction : (int -> int) -> int -> int -> int = <fun>
```

On peut lire cette ligne comme :

- `ma_fonction` prend 3 paramètres :
 1. Une fonction de type `int -> int`
 2. Une valeur de type `int`
 3. Une valeur de type `int`
- `ma_fonction` renvoie une valeur de type `int`

H Typage

On a pu remarquer depuis le début du chapitre qu'OCaml peut deviner les types des expressions, grâce à un système appelé **l'inférence de type**. Écrivons le code de la fonction *identité*, qui renvoie son entrée :

```
1 let f x = x;;
```

Le type de cette fonction est `'a -> 'a`. Cette notation un peu particulière veut dire : “pour tout type α , on peut donner à la fonction le type $\alpha \rightarrow \alpha$ ”. Contrairement au C, en OCaml, une fonction peut prendre et renvoyer plusieurs types totalement différents. Par exemple, la fonction identité peut prendre n'importe quel type et renvoie une valeur du même type que son entrée.

Définition 7

On dit qu'une fonction est **polymorphe** si elle peut prendre en entrée et renvoyer plusieurs types.

On dit qu'OCaml utilise le *polymorphisme* de type.

Introduisons une nouvelle syntaxe : `(x1, x2, ..., xn)`. En OCaml, comme en Python, on peut créer des **tuples**. Regardons le type des tuples :

Q14. Taper les expressions suivantes et regarder leur type :

```
1 let p1 = (1, 2, 3) ;;
2 let p2 = (1, 'a', "toto", 4);;
3 let diago x = (x, x);;
```

Si e_1, e_2, \dots, e_n sont des expressions de types respectifs `t1, t2, ..., tn`, alors le tuple (e_1, e_2, \dots, e_n) est de type `t1 * t2 * ... * tn`. On dit que c'est un type **produit**.

Pour utiliser un tuple, il est nécessaire de le **déstructurer**. Cela signifie que l'on va extraire du tuple les différentes composantes. Par exemple :

```
1 let p1 = (1, 2, 3) ;;
2 let (x, y, z) = p1;;
```

On a **déstructuré** `p1` en utilisant un tuple de variables ayant la même **structure** : `(x, y, z)`. Lorsque l'on déstructure ainsi un tuple, ou n'importe quelle autre type (cf plus loin), on parle de **let déstructurant**.

On peut également déstructurer des tuples directement dans les paramètres d'une fonction. Par exemple, les deux fonctions suivantes sont équivalentes :

```
1 let echange1 p =  
2   let (x, y) = p in (y, x)  
3  
4 let echange2 (x, y) = (y, x)
```

Il faut bien noter que les deux fonctions ont la même signature : `('a * 'b) -> ('b * 'a)`. Ces deux fonctions prennent donc un seul argument, de type `('a * 'b)`.

I Récapitulatif

Pour faire des commentaires en OCaml, on les entoure par `(*)`. Par exemple :

```
1 (* produit de x et y *)
2 let mul x y = x * y
```

Revoyons toutes les notions vues jusqu'ici :

- Les types de base : `int`, `float`, `bool`, `char`, `string`, `unit`
- Les opérations sur ces types : comme en C. Le **non** booléen se note `not` et les opérateurs flottants nécessitent un point : `+.` au lieu de `+` et idem pour les autres.

Type t	Opérateurs binaires pour t	Opérateurs unaires pour t	Type de l'expression composée
<code>int</code>	<code>+</code> , <code>-</code> , <code>*</code> , <code>/</code> , <code>mod</code>	<code>-</code>	<code>int</code>
<code>float</code>	<code>+</code> , <code>-</code> , <code>*</code> , <code>/</code> , <code>**</code> (puissance)	<code>-.</code>	<code>float</code>
Tous	<code>></code> , <code>>=</code> , <code>=</code> , <code><</code> , <code><=</code> , <code><></code> ("différent de")		<code>bool</code>
<code>bool</code>	<code>&&</code> , <code> </code>	<code>not</code>	<code>bool</code>
<code>char</code>			
<code>string</code>	<code>^</code> (concaténation)		<code>string</code>

- On peut ajouter une variable au contexte globale avec `let x = ... ;;` et ajouter une variable dans un contexte local avec `let x = ... in ...`
- On peut définir des fonctions avec la syntaxe suivante :

```
1 let f x1 x2 ... xn =
2   ...
```

- On applique une fonction avec :

```
1 f e1 e2 ... ek
```

où `e1, e2, ... ek` sont des expressions ayant des types compatibles avec la signature de la fonction `f`.

- Les parenthèses servent à encadrer les sous-expressions **et pas à appliquer les fonctions**

Voici un exemple de programme mettant en oeuvre ces différentes notions :

```
1 let x = 3;;
2 let y = 5;;
3 let z = 3 * 5 ;;
4
5 (* composée de f et g, i.e. la fonction h
6   telle que h(x) = f(g(x)) *)
7 let composition f g =
8   fun x -> f (g x);;
9
10 (* double de x *)
11 let double x = 2*x;;
12
13 let u = composition double int_of_string ;;
14 let a = u "12";; (* vaut 24 *)
15
16 let quadrupler = composition double double;;
17 let b = quadrupler 5;; (* vaut 20 *)
```

Quiz Pour chacune des expressions suivantes, prédire son type et sa valeur :

```
1 (* Q1 *)
2 let g x = 2*x in
3 let f x = g x + 1 in
4 f 5 + f 3;;
5
6
7 (* Q2 *)
8 let double x = 2*x in
9 let triple x = 3*x in
10 double (triple 5);;
11
12 (* Q3 *)
13 let s = "bonjour " in
14 let saluer x = s ^ x ^ " !" in
15 let s = "salut " in
16 saluer "Jérémy";;
17
18 (* Q4 *)
19 let f x y z = (x z) (y z) in
20 let g x = let x = x - 1 in x * x in
21 let h x = fun y -> let x = y in x+1 in
22 f h g 3;;
23
24 (* Q5 *)
25 let u f (x, y) = f x y in
26 let g = u (fun x -> (fun y -> x y)) in
27 g ((fun a -> a+3), 5);;
```

2 Éléments de base d'OCaml

Étudions plusieurs éléments du langage OCaml. Pour chaque nouvel élément, on regardera :

- Un exemple simple ;
- La syntaxe ;
- Le typage, c'est à dire les règles à appliquer pour typer l'expression ;
- La sémantique, c'est à dire la manière dont on évalue la valeur de l'élément.

A If-then else

Comme tous les langages, OCaml dispose d'une construction *if-then-else* :

```

1 (* Renvoie le maximum entre x et y *)
2 let maximum x y =
3   if x < y then y
4   else x
5
6 (* applique un ET logique sur a et b deux booléens*)
7 let logic_and a b =
8   if a then b else false

```

La syntaxe est la suivante :

```

1 if E1 then E2 else E3

```

où E1, E2 et E3 sont des expressions.

Pour le typage, l'expression $E1$ doit être du type `bool`, et les expressions $E2$ et $E3$ doivent être du même type T , et alors l'expression totale est de type T également.

Pour la sémantique, voici comment évaluer la valeur de `if E1 then E2 else E3` :

1. Évaluer E1 en une valeur v_1
2. Si v_1 est `true`, évaluer E2
3. Sinon, évaluer E3

Comme les deux branches du if-else doivent avoir le même type, on ne peut pas avoir de if sans else. Une exception : le type `unit` :

```

1 if true then ();;
2 if x < y then print_int x;;

```


B Matching

Les mots-clés *match* et *with* permettent de faire des disjonctions de cas plus puissantes qu'un simple if-else. Voyons quelques exemple :

```

1 (* Renvoie le nom de x si c'est un chiffre *)
2 let nom x = match x with
3   | 0 -> "Zero"
4   | 1 -> "Un"
5   | ...
6   | 9 -> "Neuf"
7
8 (* renvoie true si x vaut 0, false sinon *)
9 let is_zero x = match x with
10  | 0 -> true (* si x vaut 0, alors true *)
11  | y -> false (* si x vaut n'importe quoi d'autre, alors false *)
12
13
14 (* Première position d'un 0 du triplet t. -1 si t n'a aucun 0 *)
15 let zero_pos t = match t with
16  | (0, y, z) -> 0 (* si la 1ere composante vaut 0 *)
17  | (x, 0, z) -> 1 (* si la 2eme composante vaut 0 *)
18  | (x, y, 0) -> 2 (* si la 3eme composante vaut 0 *)
19  | u -> -1 (* Tous les autres cas *)
20
21 (* Renvoie le produit des composantes de p si aucune n'est nulle, sinon la somme *)
22 let prod_or_sum p = match p with
23  | (0, y, z) -> y + z
24  | (x, 0, z) -> x + z
25  | (x, y, 0) -> x + y
26  | u -> x * y * z

```

La syntaxe précise de cette construction est :

```

1 match E with
2 | M1 -> E1
3 ...
4 | Mn -> En

```

où E, E1, ... En sont des expressions, et où M1, ..., Mn est un **motif**.

Définition 8

Un **motif** est (définition temporaire) :

- Soit une constante
- soit une variable
- soit un tuple de motifs

Toutes les variables d'un motif doivent être distinctes.

Un motif représente un **squelette de valeur**. Par exemple, \boxed{x} est un motif qui veut dire “n'importe quelle valeur”, et $\boxed{(x, (y, 0), z, (t, u, 5))}$ est un motif signifiant “un quadruplet, dont :

1. le premier membre est n'importe quoi,
2. le deuxième est un couple dont la deuxième composante est nulle
3. le troisième est n'importe quoi
4. le quatrième est un triplet dont la troisième composante est 5

Ainsi, si l'on compare ce dernier motif avec `(3, (2, 0), 3, (7, 8, 0))`, le motif et la valeur correspondent, et alors `x` prend la valeur 3, `y` prend la valeur 2, etc...

Attention, les termes suivants ne sont pas des motifs :

```
1 | 1+x (* pas d'opérateur autorisé *)
2 | (x, x) (* pas de variable en double *)
```

Le typage d'un match with est comme suit : E peut être de n'importe quel type, du moment que le type respecte les motifs M1, ... Mn. Les E1, ..., En doivent être d'un même type T, et alors l'expression totale est de type T également. Pour chaque couple Mk, Ek, il faut que les types des variables de Mk soient cohérents dans Ek.

La sémantique de cette construction est comme suit. Pour évaluer l'expression :

```
1 match E with
2 | M1 -> E1
3 ...
4 | Mn -> En
```

il faut :

1. Évaluer E en une valeur *v*
2. Comparer *v* avec les motifs M1, ..., Mn, dans l'ordre
3. Lorsqu'un motif Mk correspond, on évalue l'expression Ek correspondante, en ayant ajouté au contexte les variables du motif Mk en les faisant correspondre à *v*.

Exemple 3

Voyons en détail comme est évaluée l'expression suivante :

```
1 match (1, (2, 1+2), 3+1) with
2 | (0, _, _) -> 0
3 | (x, (y, 0), z) -> (x - y) * z
4 | (x, (y, _), z) -> (x + y) * z
```

La variable `_`, appelée “underscore”, joue un rôle particulier : elle peut être présente plusieurs fois dans un motif, mais ne figurera pas dans le contexte. On appelle ce motif particulier un *joker*, ou *wildcard* en anglais, il sert donc à ignorer ou à jeter à la poubelle des parties de la valeur matchée. Par exemple :

```
1 (* Calcule x*y *)
2 let mult x y = match x, y with
3 | 0, _ -> 0
4 | _, 0 -> 0
5 | _ -> x*y
```

Variables et motifs Un motif est un squelette, il décrit donc une **forme** mais ne contient pas de valeur. Par exemple :

```
1 let equal x y =
2   match x with
3   | y -> true
4   | _ -> false
```

La fonction ci-dessus n'est pas correcte. En effet, le `y` dans le premier motif n'a aucun lien avec le `y` en paramètre. On pourrait remplacer le `y` du motif par n'importe quel identifiant, et même par un underscore :

```
1 let equal x y =
2   match x with
3   | _ -> true
4   | _ -> false
```

Il est alors clair que cette fonction renvoie **toujours** true !

A retenir : on ne peut **pas** utiliser une variable préexistante dans un motif pour comparer la valeur du match à la valeur de cette variable.

Matching incomplet Lorsqu'un match with ne couvre pas tous les cas possibles du type concerné, on dit qu'il est incomplet, ou non-exhaustif. Par exemple :

```
1 let f x = match x with
2   | 0 -> 1
3   | 2 -> 3
```

Si l'on essaie d'évaluer cette expression, OCaml va raler, et dire que le matching n'est pas exhaustif. Il donnera même un exemple de valeur qui n'est pas matchée ! On doit **toujours** couvrir tous les cas dans un match with. Cependant, il se peut qu'un cas ne soit pas sensé arriver car on l'empêche dans le code. Dans ce cas, on peut utiliser la fonction `failwith`, qui affiche un message d'erreur et arrête le programme. Par exemple :

```
1 let parite x = x mod 2
2 let est_pair y = match parite y with
3   | 0 -> true
4   | 1 -> false
5   | _ -> failwith "Ne doit pas arriver"
```

Il ne faut **jamais** laisser un matching incomplet.

On peut aussi utiliser les motifs avec les let-in. On parle alors de **let destructurant**. On peut aussi utiliser le joker dans ce contexte :

```
1 let p = (2, "bla", true)
2 let a, b, c = p
3
4 let p = "important", "a jeter", "aussi a jeter"
5
6 let x, _, _ = p
```

C Fonctions récursives

En OCaml, on n'utilise pas (pour l'instant) de boucle `for` ou de boucle `while`. A la place, on utilise des **fonctions récursives**, c'est à dire des fonctions qui s'appellent elles-mêmes. La récursivité est au coeur de la programmation fonctionnelle.

Par exemple, écrivons une fonction qui calcule a^b pour $a, b \in \mathbb{N}$. Par définition :

$$\forall a \in \mathbb{N}, \forall b \in \mathbb{N}, a^b = \prod_{i=1}^b a$$

Cependant, on peut remarquer la chose suivante pour $a, b \in \mathbb{N}$:

- Si $b = 0$, $a^b = 1$
- Sinon, $a^b = a^{b-1} \times a$

Ces relations permettent de **caractériser** ou même de **définir** récursivement la quantité a^b . On voudrait donc écrire :

```
1 let puissance a b =
2   if b = 0 then 1
3   else a * puissance a (b-1)
```

mais cela n'est pas accepté. En effet, au moment où l'on écrit le corps de la fonction `puissance`, l'identifiant `puissance` ne fait pas partie de l'environnement. En OCaml, pour autoriser une fonction à utiliser son propre nom, i.e. pour préciser que l'on écrit une fonction récursive, on utilise le mot clé **let rec** :

```
1 let rec puissance a b =
2   if b = 0 then 1
3   else a * puissance a (b-1)
```

Le typage et la sémantique du `let rec` sont les mêmes que pour le **let** classique, mais à l'évaluation, une fonction récursive se trouve dans son propre contexte.

Exercice 4

Créez un fichier “exp.ml”. Vous pourrez l'exécuter dans l'interpréteur en y tapant :

```
1 #use "exp.ml";;
```

Q1. Recopiez la fonction `puissance`. Exécutez le code dans l'interpréteur puis testez la fonction sur quelques exemples.

Q2. Trouver une formule de récurrence pour la factorielle, et en déduire une fonction OCaml `factorielle: int -> int`.

Q3. Trouver une formule de récurrence permettant d'écrire l'exponentiation rapide sous forme récursive :

- Si $b = 0$, `exp a b` = ...
- Sinon :
 - Si b est pair, `exp a b` = ... (faire apparaître un terme de la forme `exp X Y`)
 - Sinon, `exp a b` = ... (faire apparaître un terme de la forme `exp X Y`)

Q4. Implémenter l'exponentiation rapide, ré-exécuter le fichier “exp.ml” puis tester la fonction.

D Listes

Un nouveau type : les listes. En OCaml, les listes sont définies récursivement, comme suit :

- La liste vide, notée `[]`, est une liste
- Si $E1$ est une expression de type `'a` et $E2$ une expression de type `'a list`, alors `E1::E2` est une liste, contenant $E1$ suivi des éléments de $E2$. On dit que $E1$ est la *tête* de liste, et que $E2$ est la *queue* de liste.

Attention, la tête d'une liste est un élément, mais la queue d'une liste est une liste. On peut donc voir les listes OCaml comme des piles : on n'a accès qu'à l'élément au début d'une liste.

Exemple 4

Voyons des exemples de listes :

```
1 let l_vide = []
2 let l_simple = 3::6::8::[] (* comme 3 :: (6 :: (8 :: [])) *)
3
4 (* Renvoie une liste de taille n contenant
5    exclusivement des 1 *)
6 let rec ones n = match n with
7   | 0 -> []
8   | _ -> 1::ones (n-1)
9
10 let cinq_uns = ones 5
```

On remarque qu'OCaml affiche les listes sous la forme `[v1; v2; ... ; vn]`. On peut également utiliser cette syntaxe pour définir des listes :

```
1 let l_a = [1;2;3;4;5;6]
2 let l_b = 1::2::3::4::5::6::[] ;
```

La première version est du sucre syntaxique, c'est la deuxième forme qui est plus fidèle à la forme qu'ont les listes.

Une liste doit contenir uniquement des expressions d'un même type T , et dans ce cas, la liste sera de type `T list`. Par exemple, `["blabla"; "bli"; "toto"]` est de type `string list`, mais `[1; 2.1; 0]` contient des entiers et des flottants, et n'est donc pas une expression bien typée.

Les listes forment un **type polymorphique**. Par exemple, la liste vide est typée `'a list` par OCaml. Cela signifie que c'est une liste de type `T list`, pour tout type T . Ainsi, lorsque l'on écrira des fonctions générales sur les listes, on pourra les appliquer sur des listes de n'importe quel type !

Les listes viennent également agrandir la liste des *motifs* :

- `[]` est un motif
- si `M` et `L` sont des motifs, alors `M::L` est un motif

Par exemple : `x::y::q` est un motif signifiant "deux éléments puis une liste". Ce motif sera compatible avec toutes les listes de taille au moins 2, et permettra de récupérer les deux premiers éléments, et la liste des éléments suivants. `(x, y) :: q` est un motif signifiant "une liste de couples, d'au moins un élément". Il permettra de matcher par exemple `[("bla", 5); ("titi", 129)]`, et alors `x` vaudra `"bla"` et `y` vaudra `5`.

Utilisons ces motifs pour créer quelques fonctions :

```

1 (* Renvoie la somme des éléments de l: int list *)
2 let rec somme_liste l = match l with
3   | [] -> 0
4   | x::q -> x + somme_liste q
5
6 (* Renvoie true si la liste est de taille pair, false sinon *)
7 let rec taille_paire l = match l with
8   | [] -> true
9   | x::[] -> false
10  | x::y::q -> taille_paire q (* enlever deux éléments conserve la parité *)

```

Tri par sélection Écrivons le tri par sélection en OCaml. On aura besoin d'une fonction de sélection qui extrait le max d'une liste : codons une fonction `selection: 'a list -> ('a * 'a list)` qui étant donné une liste L renvoie le couple (x, L') avec x le max de L , et L' la liste L où une occurrence de x a été supprimée.

Pour coder cette fonction, on doit réfléchir inductivement : si j'ai une liste Q et que j'ai obtenu son max m et la liste Q' des autres éléments, comment faire pour obtenir le max de $x :: Q$? Il suffit de regarder le maximum entre x et m : c'est le maximum de $x :: Q$, et l'autre peut être remis au début de la liste. Cette remarque nous donne une définition inductive de la fonction :

```

1 (* Couple (a, t) avec a le maximum de l et t la
2   liste l privée d'une occurrence de a *)
3 let rec selection l = match l with
4   | [] -> failwith "empty list"
5   | x::[] -> x, []
6   | x::q -> let (m, q') = selection q in
7             if x > m then (x, m::q')
8             else (m, x::q')

```

A partir de cette fonction, on peut construire aisément une fonction récursive de tri par sélection : si la liste n'est pas vide, on extrait son maximum, on trie les éléments restants et on rajoute l'élément extrait au début.

```

1 let rec select_sort l = match l with
2   | [] -> []
3   | x::q -> let (m, q) = selection l in m::(select_sort q)

```

E Alias de type

En OCaml, on peut définir ses propres types. Par exemple, comme en C, on peut redéfinir un type pré-existant en lui donnant un autre nom :

```
1 type nombre = int
2 type vecteur3d = float * float * float
3
4 (* fonctions |N -> |B *)
5 type filtre_entier = int -> bool
```

Bien qu'OCaml devine tout seul le type des expressions que l'on écrit, il ne devine pas toujours par lui même les types définis ainsi :

```
1 let x = (0., 0., 0.) (* affiche float*float*float *)
2 let y = 5 (* affiche nombre plutôt que int *)
```

Cependant, on peut spécifier le type des variables avec “:” comme suit :

```
1 let x: int = 5
2
3 let f (x:int) : int = x + 1 (* prend en entrée un int et renvoie un int *)
4
5 let f: (int -> int) = fun x -> x - 1
6
7 let p: vecteur3d = (2., 3.2, -8.54)
8
9 let equal_5: filtre_entier = fun x -> x = 5
10
11 let produit_scalaire (a:vecteur3d) (b:vecteur3d) : float =
12   let xa, ya, za = a in
13   let xb, yb, zb = b in
14   xa *. xb +. ya *. yb +. za *. zb
15
16
17 (* Ici, le type vecteur3d serait inféré automatiquement à cause du type
18    de produit_scalaire *)
19 let norme_carree (a:vecteur3d) : float = produit_scalaire a a
```

F Type somme

Le type produit correspond au produit cartésien d'ensemble. Le type somme correspond plutôt à l'union disjointe.

Un type somme permet de représenter des catégories d'objets ayant plusieurs “cas”. Un tel type est constitué de **constructeurs**, qui sont les différents mots clés que l'on utilise pour construire des objets de ce type. Chaque constructeur correspond à un “cas” différent.

Exemple 5

On veut implémenter un type pour les fournitures scolaires. On veut pouvoir représenter :

- Les stylos BIC, qui peuvent être de couleurs différentes
- Les règles, qui peuvent être de tailles différentes et peuvent être centrées ou pas (i.e. 0 est au centre ou au bord)
- Les gommes

Créez un fichier “fourniture.ml”. Vous pourrez l’exécuter dans l’interpréteur en tapant :

```
1 #use "fourniture.ml";;
```

On définit le type somme avec :

```
1 type fourniture =
2   | Stylo of string (* couleur *)
3   | Regle of int * bool (* taille en cm, centrée ou non *)
4   | Gomme
```

Ce qui se lit : “Il y a trois types de fournitures : Les stylos, qui sont paramétrés par une chaîne de caractères, les règles, paramétrées par un entier et un booléen, et les gommes”. Les commentaires précisent à quoi servent les paramètres.

On peut ensuite créer des éléments de ce type :

```
1 let x = Stylo "rouge"
2 let r1 = Regle (30, true)
3 let r2 = Regle (20, false)
4 let g = Gomme
```

Notons qu’OCaml peut automatiquement tester l’égalité entre deux objets d’un même type :

```
1 let x = Stylo "rouge"
2 let y = Stylo "rouge";;
3 x = y;; (* Vaut true *)
```

La manière principale de manipuler les types définis ainsi est le *match with*. Par exemple, on suppose que le prix des fournitures est comme suit :

- Les gommes coûtent 1,50€;
- Les stylos bleus coûtent 1,20€, les autres coûtent 1€;
- Une règle de l centimètres coûte $1 + \frac{l}{15}$ euros.

Voici comment on implémenterait une fonction calculant le prix d’une fourniture en OCaml :

```
1 (* prix de fourn en euros *)
2 let prix (fourn: fourniture) : float =
3   match fourn with
4   | Gomme -> 1.5
5   | Stylo "bleu" -> 1.2
6   | Stylo _ -> 1.0
7   | Regle(longueur, _) -> 1.0 +. float_of_int longueur /. 15.0
```

On peut représenter une trousse comme une liste de fournitures. Écrivons une fonction qui calcule le nombre de gommes dans une trousse :

```
1 type trousse = fourniture list;;
2
3 let rec nombre_gommes (t:trousse) : int =
4   match t with
5   | [] -> 0
6   | Gomme::q -> 1 + nombre_gommes q
7   | _::q -> nombre_gommes q
```


Exercice 5

On suppose qu'une trousse seule coûte 5€. Écrire une fonction calculant le prix total d'une trousse, en comptant tout ce qu'elle contient :

```
1 let rec prix_trousse (t: trousse) : float = ...
```

La syntaxe pour les type somme est donc ;

```
1 type nom_type =
2   Constructeur1 of type1 (* ou Constructeur1 *)
3   | Constructeur2 of type2 (* ou Constructeur2 *)
4   ...
5   | ConstructeurN of typeN (* ou ConstructeurN *)
```

On rajoute également à la définition des *motifs* :

- Si `Cons` est un constructeur d'un type somme, et `M` est un motif, alors `Cons M` est un motif.

Rien n'empêche un type de s'auto-référencer, on dit alors que c'est un type *inductif*, ou récursif. Par exemple, si l'on veut créer un type représentant les expressions arithmétiques :

```
1 type expr =
2   | Constante of int
3   | Plus of expr * expr
4   | Foix of expr * expr
```

Autrement dit : une expression est soit une constante entière, soit la somme de deux expressions soit le produit de deux expressions. L'expression $(1 + 2)$ s'écrira :

```
1 Plus (1, 2)
```

L'expression $(1 + 2) * (3 + 7 * 6)$ s'écrira :

```
1 let my_expr =
2   Foix (
3     Plus(
4       Constante 1,
5       Constante 2
6     ),
7     Plus(
8       Constante 3,
9       Foix (
10        Constante 7,
11        Constante 6
12      )
13    )
14  )
```

Pour manipuler des types inductifs, il faudra généralement utiliser des fonctions récursives. Sur l'exemple précédent, écrivons une fonction qui évalue une expression arithmétique :

```
1 let rec eval e = match e with
2   | Constante n -> n
3   | Plus (e1, e2) -> eval e1 + eval e2
4   | Foix (e1, e2) -> eval e1 * eval e2
```

(Remarquez que cette fonction ressemble de près à l'algorithme d'évaluation des expressions OCaml décrit plus tôt dans le chapitre. En fait, à ce stade, vous pourriez créer un type pour les expressions OCaml, un type pour les environnements, et recoder OCaml en OCaml!)

Tentons de recoder le type des listes OCaml. On remarque que les listes sont polymorphes, autrement dit elles peuvent s'adapter à des types différents. On voudrait écrire :

```
1 type liste =
2   | Vide (* liste vide *)
3   | Cons of 'a * liste (* Constructeur: Cons (x, q) sera x suivi de q *);;
```

Cependant, OCaml affiche : `Error: The type variable 'a is unbound in this type declaration.`

Pour définir ce type, on doit le *paramétrer* :

```
1 type 'a liste =
2   | Vide
3   | Cons of 'a * ('a liste) ;;
```

La syntaxe générale des types paramétrés est :

```
1 type ('a1, 'a2, ..., 'an) nom_type =
2   | Constructeur1 of type1
3   | ...
4   | Constructeurk of typek
5   ;;
```

On peut ensuite manipuler ce type normalement, et OCaml gèrera le polymorphisme :

```
1 let ma_liste_1 = Cons (5, Cons(6, Vide)) ;; (* [5, 6] *)
2 let ma_liste_2 = Cons ("bla", Cons("blo", Cons ("bli", Vide))) ;; (* ["bla", "blo", "bli"] *)
3
4 let tete l = match l with
5   | Vide -> failwith "liste vide"
6   | Cons (x, q) -> x
7   ;;
```

Exercice 6

Écrire une fonction `vraie_liste: 'a liste -> 'a list` qui transforme une liste de notre type liste maison en une liste standard OCaml :

3 Récursivité

A Correction d'une fonction récursive

Lorsque l'on écrit du code OCaml, on n'utilise pas de boucles. Donc, il nous faut un nouvel outil pour prouver la correction des fonctions. Considérons un exemple simple : la multiplication :

```
1 (* Renvoie x fois y, pour x, y entiers positifs *)
2 let rec mult (x: int) (y: int) : int =
3   if x = 0 then 0
4   else y + mult (x-1) y
```

Cette fonction est correcte si elle renvoie bien ce qui est indiqué dans sa documentation, donc ici si elle renvoie bien le produit de ses deux arguments. Remarquons que le commentaire de la fonction précise une précondition nécessaire à la correction : x et y doivent être positifs.

Pour étudier cette expression, on introduit la fonction mathématique correspondante :

$$\mathbf{mult} : \mathbb{N}^2 \longrightarrow \mathbb{N} \quad x, n \mapsto \begin{cases} 0 & \text{si } n = 0 \\ x + \mathbf{mult}(x, n-1) & \text{sinon} \end{cases}$$

Cette fonction est *définie par récurrence*. Commençons par montrer qu'elle est bien définie, c'est à dire que la fonction OCaml correspondante termine. Pour cela, il faut remarquer qu'un appel à la fonction **mult** cause un appel récursif, sur une entrée strictement plus petite. Comme les entrées sont dans \mathbb{N} (par précondition), et comme il n'existe pas de suite infinie strictement décroissante dans \mathbb{N} , la fonction s'arrête bien.

Montrer la correction de la fonction, c'est montrer la propriété suivante :

$$\forall x \in \mathbb{N}, \forall n \in \mathbb{N}, \mathbf{mult}(x, n) = x \times n$$

Pour cela, on procède par récurrence. Plus précisément, posons $x \in \mathbb{N}$, et montrons par récurrence sur \mathbb{N} la propriété suivante :

$$\forall n \in \mathbb{N}, P(n) : \mathbf{mult}(x, n) = x \times n$$

- Pour $n = 0$, $\mathbf{mult}(x, 0) = 0$: $P(0)$ est vraie.
- Soit $n \in \mathbb{N}^*$, supposons $P(n-1)$. Alors, $\mathbf{mult}(x, n-1)$ vaut $x \times (n-1)$. Or, $n > 0$ donc $\mathbf{mult}(x, n) = x + \mathbf{mult}(x, n-1) = x + x \times (n-1)$ par hypothèse de récurrence. Donc, $\mathbf{mult}(x, n) = x \times n$: $P(n)$ est vraie

Voyons un exemple plus complexe, en étudiant le tri par insertion :

```

1 (* insert x l est une copie de l où x a été inséré dans l'ordre.
2   l doit être triée *)
3 let rec insert (x: 'a list) (l: 'a list) : 'a list = match l with
4 | [] -> [x]
5 | y::q -> if x < y then x :: l
6           else y :: insert x q ;;
7
8 (* insert_sort l est une copie triée de l *)
9 let rec insert_sort (l: 'a list) : 'a list = match l with
10 | [] -> []
11 | x::q -> let q = insert_sort q in insert x q ;;

```

On considère les fonctions mathématiques correspondantes. On ne décrit pas pour l'instant l'ensemble mathématique précis où vivent les listes, on y viendra au chapitre suivant.

$$\begin{aligned}
 \text{insert} & : x, L \mapsto \begin{cases} [x] & \text{si } L \text{ est vide} \\ x :: y :: Q & \text{si } L \text{ de la forme } y :: Q \text{ et } x < y \\ y :: (\text{insert}(x, Q)) & \text{si } L \text{ de la forme } y :: Q \text{ et } x \geq y \end{cases} \\
 \text{tri.insert} & : L \mapsto \begin{cases} [] & \text{si } L \text{ est vide} \\ (\text{insert}(x, \text{tri.insert}(Q))) & \text{si } L \text{ de la forme } x :: Q \end{cases}
 \end{aligned}$$

Ces fonctions sont définies par récurrence sur la taille de L . Au prochain chapitre, on dira qu'elles sont définies par *induction* sur la structure des listes.

Commençons par la terminaison d'insert. On pose $f : L \mapsto |L|$. Soit L une liste, notons L_0, \dots, L_k, \dots les listes sur lesquelles on appelle récursivement la fonction **insert** lors de l'évaluation de **insert**(x, L). On remarque que la suite $(f(L_k))$ est strictement décroissante et à valeurs dans \mathbb{N} . C'est donc une suite finie : la fonction **insert** termine.

La terminaison de la fonction de tri se prouve de manière analogue. Montrons maintenant formellement la correction des deux fonctions. Au chapitre suivant, nous verrons une généralisation du principe de récurrence qui permettra d'écrire de manière beaucoup plus élégante et courte les preuves de ce style.

— Pour **insert** : On raisonne par récurrence sur la taille de la liste L en entrée. On pose x un élément à insérer. Montrons la propriété suivante par récurrence sur $n \in \mathbb{N}$:

$\forall n \in \mathbb{N}, P(n) : “\forall L \text{ liste de taille } n, \text{ si } L \text{ est triée, alors } \text{insert}(x, L) \text{ est contient les éléments de } L \text{ ainsi que } x, \text{ et est triée.}”$

- Si $|L| = 0$ alors $L = []$, et $[x]$ contient bien les éléments de L (personne) ainsi que x , et est triée.
- Soit L une liste avec $|L| > 0$. On écrit $L = y :: Q$. Si $x < y$ alors $x :: y :: Q = x :: L$ est bien une copie de L où x a été ajouté, triée. Sinon, par hypothèse de récurrence, **insert**(x, Q) est une copie triée de Q où x a été ajouté, et comme L est triée, $y \leq \min Q$ et $y \leq x$. Donc, $y :: \text{insert}(x, Q)$ est triée, et contient bien x ainsi que les éléments de L , à savoir y et les éléments de Q

— Pour **insert_sort** : **Exercice** : Montrer par récurrence la propriété suivante :

$\forall n \in \mathbb{N}, P(n) : “\forall L \text{ liste de taille } n, \text{ insert_sort}(L) \text{ est triée et contient les éléments de } L.”$

B Récursivité terminale

On rappelle que les appels de fonctions nécessitent de réserver de la mémoire pour s'exécuter : cette mémoire est réservée dans la **pile d'appel** (voir chapitre 3), et les appels récursifs viennent s'ajouter les uns sur les autres. Dans un langage comme OCaml, où les appels récursifs sont un outil central de programmation, il faut faire attention à ne pas remplir la pile. En théorie, il faudrait donc éviter de faire des appels récursifs trop profonds (par exemple avoir un million de stack frames empilées les unes sur les autres). En pratique, certaines fonctions récursives peuvent être transformées en boucle par le compilateur, on peut donc les exécuter sans se soucier de problèmes de dépassement de pile. Voyons un exemple : considérons les deux fonctions suivantes, qui permettent de calculer la factorielle.

```

1 (* factorielle n = n! *)
2 let rec factorielle n =
3   if n = 0 then 1
4   else n * factorielle (n-1)
5
6 (* factorielle_bis n accu = n! × accu *)
7 let rec factorielle_bis n accu =
8   if n = 0 then accu
9   else factorielle_bis (n-1) (accu*n)
10
11 (* On remarque que factorielle_bis n 1 = factorielle n *)

```

Pour évaluer `factorielle 2`, on doit :

1. Calculer factorielle 2 :
 - (a) Calculer factorielle 1
 - i. Calculer factorielle 0
 - ii. On renvoie donc 1
 - (b) On obtient 1, on multiplie par 1, on renvoie donc 1
2. On obtient 1, on multiplie par 2, on renvoie donc 2

Ainsi, on doit plonger au fond des appels récursifs, puis remonter à la surface en appliquant les opérations nécessaires. On doit donc **empiler** les appels de fonction les un sur les autres, puis les **dépiler**. Donc, si l'on tente d'évaluer `factorielle 10000000`, on aura une erreur de dépassement de pile.

En revanche, pour évaluer `factorielle_bis 2 1` :

1. Calculer factorielle_bis 2 1
 - (a) Calculer factorielle_bis 1 2
 - i. Calculer factorielle_bis 2 0
 - ii. On renvoie donc 2
 - On renvoie donc 2
- On renvoie donc 2

Ici, la phase de “remontée à la surface” est triviale : elle sert juste à transmettre le résultat vers l'appel initial. Autrement dit, on n'a pas besoin de sauvegarder les différentes stack frames dans la pile, car on n'utilise plus leurs données une fois qu'on est entré dans un appel plus profond. Les différentes stack frames peuvent donc se remplacer les unes les autres, au lieu de s'empiler. Les compilateurs d'OCaml peuvent détecter les fonctions ayant cette propriété, et les transformer en boucles dans le code machine généré. Si l'on évalue `factorielle_bis 10000000 1`, aucun problème de pile !

Définition 9

On appelle fonction **récursive terminale** (en anglais : **tail-recursive**) toute fonction récursive ne nécessitant aucun traitement à la remontée d'une valeur.

Proposition 1

Les fonctions récursives terminales peuvent être compilées / interprétées de façon à garder une taille de pile d'appel constante.

Remarquons que pour `factorielle_bis`, on n'écrit pas directement la fonction factorielle, mais une fonction **plus générale**, que l'on applique avec un bon paramètre ensuite. La programmation récursive terminale nécessite souvent d'utiliser une méthode par **accumulateur** qui consiste à stocker le résultat temporaire dans un ou plusieurs des paramètres, appelés les accumulateurs.

Exemple 6

Prenons la fonction d'exponentiation rapide et tentons d'en trouver une version récursive terminale.

```
1 let rec fast_exp a b =
2   match (b, b mod 2) with
3   | (0, _) -> 1
4   | (_, 0) -> fast_exp (a*a) (b/2)
5   | _      -> a * fast_exp (a*a) (b/2) (* Pas récursif terminal ! *)
```

On rappelle la version impérative de l'exponentiation rapide :

```
1 int fast_exp(int a, int b){
2   int r = 1, A = a, B = b;
3   while (B>0){ // invariant:  $r \times A^B = a^b$ 
4     if (B%2 == 1){
5       r = A*r;
6     }
7     a = A*A;
8     b = B/2;
9   }
10  return r;
11 }
```

Nous allons utiliser l'invariant pour déterminer une fonction récursive terminale. Pour $A, B, r \in \mathbb{N}$, notons $f(A, B, r) = r \times A^B$. L'invariant de boucle peut se réécrire : $f(A, B, r) = f(a, b, 1) = a^b$. Autrement dit, la quantité $f(A, B, r)$ reste constante au long de l'exécution, et sa valeur est a^b . Il nous suffit donc de déterminer une formule de récurrence permettant de calculer f , ce que l'on obtient en traduisant le code de la boucle while :

$$\begin{aligned} f(A, 0, r) &= r && \text{(sortie de boucle)} \\ f(A, B, r) &= f(A^2, \frac{B}{2}, r) && \text{si } B \text{ pair} \\ f(A, B, r) &= f(A^2, \frac{B-1}{2}, Ar) && \text{si } B \text{ impair} \end{aligned}$$

Ces formules nous donnent immédiatement une version récursive terminale :

```
1 (* exp_mult a b r =  $r \times a^b$  *)
2 let rec exp_mult a b r =
3   match (b, b mod 2) with
4   | (0, _) -> r
```

```

5 | (_, 0) -> exp_mult (a*a) (b/2) r
6 | _      -> exp_mult (a*a) (b/2) (a*r)
7
8 (* exp a b = a^b *)
9 let exp a b =
10   exp_mult a b 1

```

La variable `r` du code OCaml va jouer le rôle de `r` dans le code C et grossir en stockant de plus en plus d'information, pour être renvoyée à la fin. On appelle un tel paramètre de fonction un **accumulateur**.

Dans l'exemple précédent, nous sommes repartis d'une boucle pour trouver une version récursive terminale, mais il est plus simple de réfléchir directement à comment introduire un accumulateur : que stocker dedans, et comment l'initialiser.

Règles de bonne conduite Lorsque l'on programme en OCaml, on essaie d'écrire des fonctions récursives terminales dès que c'est possible, car elles sont plus efficaces d'un point de vue mémoire. On passe alors quasi-systématiquement par des fonctions auxiliaires utilisant un accumulateur, et il y a alors deux possibilités :

- On définit la fonction auxiliaire **avant** la fonction principale :

```

1 let rec fonction_auxiliaire x_1 ... x_n accu_1 ... accu_m =
2   ...
3 let fonction_principale x_1 ... x_n =
4   fonction_auxiliaire x_1 ... x_n v_1 ... v_m

```

L'avantage principal est que l'on peut utiliser la fonction auxiliaire dans le reste du code (en particulier pour tester / débbugger).

- On définit la fonction auxiliaire **dans** la fonction principale :

```

1 let fonction_principale x_1 ... x_n =
2   let rec fonction_auxiliaire x_1 ... x_n accu_1 ... accu_m =
3     ...
4   in
5   fonction_auxiliaire x_1 ... x_n v_1 ... v_m

```

L'avantage principal est que l'on n'a pas besoin de re-préciser en paramètre de la fonction auxiliaire les paramètres de la fonction principale qui restent constant entre les appels. Par exemple :

```

1 (* somme f n =  $\sum_{i=0}^{n-1} f(i)$  *)
2 let somme (f: int->int) (n: int) : int =
3   (* somme_plus (k: int) (r: int) =  $r + \sum_{i=0}^{k-1} f(i)$  *)
4   let rec somme_plus k r =
5     if k < 0 then r else
6     somme_plus (k-1) (r + f (k-1))
7   in
8   somme_plus n 0

```

La fonction f étant toujours la même, pas besoin de la passer en paramètre de la fonction auxiliaire, ce qui rend le code plus léger.

Dans les deux cas, il est important de respecter certaines règles.

- Ne **jamais** appeler la fonction auxiliaire avec le même nom que la fonction principale suivi de “_aux”. Si vous cherchez des exos / cours en ligne, vous verrez souvent cette pratique, mais c'est peu recommandable.

- Autant que possible, trouver un commentaire de documentation adapté pour la fonction auxiliaire, et pas juste `(* c'est la fonction auxiliaire *)`. Ceci nécessite de réfléchir au rôle précis de l'accumulateur, et donc de donner une spécification mathématique de la fonction auxiliaire.
- Éviter de donner le même nom aux paramètres de la fonction auxiliaire et à ceux de la fonction principale si l'on code en "imbriqué".

Les deux premières règles sont liées, et sont là pour vous forcer à réfléchir au **sens** de vos fonctions auxiliaires au delà de "j'utilise un accumulateur". Par exemple pour la fonction `somme_plus` de l'exemple précédent, la spécification est claire : la fonction renvoie la **somme** des k premières valeurs de f , **plus** x . Il n'est pas toujours évident de trouver un nom adapté, mais en général, réfléchir au nom revient à réfléchir au sens, et donc à la manière d'écrire la fonction. Ces pratiques vous aideront donc sur le long terme à coder efficacement.

Exercice 7

Écrivons quelques fonctions récursives sur les listes, en cherchant systématiquement une version récursive terminale.

- Q1.** Définir une fonction `longueur` permettant de calculer la longueur d'une liste.
- Q2.** Définir une fonction `reverse` permettant de renverser une liste.
- Q3.** La fonction `map: ('a -> 'b) -> 'a list -> 'b list` qui applique une fonction à tous les éléments d'une liste.
- Q4.** Une fonction `fibonacci: int -> int` calculant les termes de la suite de Fibonacci.

Exercice 8

On considère le type suivant représentant des expressions arithmétiques :

```
1 type expr =
2   | Int of int
3   | Plus of expr * expr ;;
```

- Q15.** Définir une fonction `eval: expr -> int` qui évalue une expression. Essayer de la rendre récursive terminale. Que remarque-t'on ?


C Arbre d'appel

En OCaml, on peut *tracer* une fonction pour observer ses valeurs d'entrée et de sortie au cours des appels récursifs générés par une exécution. Par exemple :

<pre>1 let rec fact n = 2 if n <= 1 then 1 3 else n * fact (n-1) 4) 5 ;; 6 #trace fact ;; 7 fact 3;;</pre>	<pre>fact <-- 3 fact <-- 2 fact <-- 1 fact --> 1 fact --> 2 fact --> 6</pre>
--	--

$f \leftarrow x$ signifie que l'on appelle f sur x , et $f \rightarrow x$ signifie qu'un appel s'est terminé et a renvoyé x .

Un **arbre d'appel** permet de représenter schématiquement une telle exécution. On suit la trace de la fonction, en descendant dans l'arbre lors d'un nouvel appel, et en remontant lorsqu'un appel termine. Par exemple, l'arbre d'appel de `fact 4` est :



Vocabulaire : Les cercles représentant des appels sont nommés **noeuds** de l'arbre. Le premier noeud, correspondant à l'appel initial, est appelé **racine**.

L'arbre d'appel permet d'estimer la complexité d'une fonction : Si chaque appel demande un nombre constant d'opérations hors appels récursifs, la fonction aura une complexité proportionnelle au nombre de noeuds de l'arbre d'appel.

Dans l'exemple précédent, l'arbre d'appel de `fact n` a n noeuds pour $n \in \mathbb{N}^*$, et chaque appel demande un temps constant excepté l'appel récursif : la complexité est en $\mathcal{O}(n)$.

Exercice 9

On considère la suite suivante :

$$\begin{aligned} u_0 &= 2 \\ \forall n \in \mathbb{N}, u_{n+1} &= |1 - 3u_n + 2u_n^2| \end{aligned}$$

On propose la fonction suivante permettant de calculer la suite u :

```

1 let rec u n =
2   match n with
3   | 0 -> 2
4   | _ -> if 1 - 3 * u (n-1) + 2 * u (n-1) * u (n-1) > 0 then
5         1 - 3 * u (n-1) + 2 * u (n-1) * u (n-1)
6       else
7         -(1 - 3 * u (n-1) + 2 * u (n-1) * u (n-1))

```

Dessiner les quelques premiers étages de l'arbre d'appel de `u 5`. Estimer le nombre de noeuds dans l'arbre d'appel de `u n` pour $n \in \mathbb{N}$, et en déduire la complexité de cette fonction. Proposer une fonction plus rapide.

D Complexité

L'exemple précédent montre qu'il est possible de déterminer la complexité d'une fonction récursive en étudiant son arbre d'appel. Une autre méthode très générale est de trouver une relation de récurrence vérifiée par la complexité, et de la résoudre.

Exemple 7

Pour la fonction `fact`, notons C_n le coût de l'appel `fact n`. Pour $n > 1$ on a $C_n = C_{n-1} + \mathcal{O}(1)$. Donc, à partir d'un certain rang n_0 , pour une certaine constante $A > 0$, $C_n \leq C_{n-1} + A$. On a donc :

$$\begin{aligned}
C_n &\leq C_{n-1} + A \\
&\leq C_{n-1} + C_{n-2} + 2A \\
&\leq C_{n_0} + (n - n_0)A \\
&\leq C_{n_0} + An
\end{aligned}$$

La fonction est donc en $\mathcal{O}(n)$: même coût que l'algorithme avec une boucle.

Exercice 10

De même, trouver une relation de récurrence pour la fonction $\boxed{\text{u}}$ de l'exemple précédent, et retrouver la complexité asymptotique.

On considère l'exponentiation rapide, que l'on peut écrire en OCaml comme suit :

```

1 (* calcule a puissance n *)
2 let rec fast_exp (a : int) (n : int) : int = match n with
3   | 0 -> 1
4   | _ -> if n mod 2 = 0 then fast_exp (a*a) (n/2)
5           else a * fast_exp (a*a) (n/2)

```

On peut étudier la complexité C_n de cette fonction grâce à l'équation qu'elle vérifie :

$$C_n = C_{\lfloor \frac{n}{2} \rfloor} + \mathcal{O}(1)$$

Étudions pour simplifier la relation suivante :

$$\begin{aligned}
C_0 &= 1 \\
C_n &= C_{\lfloor \frac{n}{2} \rfloor} + 1
\end{aligned}$$

qui ne changera pas le comportement asymptotique. En raisonnant en base 2, on remarque que l'opération $n \mapsto \lfloor \frac{n}{2} \rfloor$ consiste à enlever un bit à n . Ainsi, on peut appliquer la relation de récurrence au plus $\lceil \log_2 n \rceil$ fois : une fois par bit dans n . On obtient alors que $C_n = 1 + 1 + \dots + 1$ avec $\lceil \log_2 n \rceil$ 1s, i.e. $C_n = \lceil \log_2 n \rceil$.

Ainsi, $C_n = \Theta(\log n)$: on retrouve la complexité logarithmique de l'exponentiation rapide impérative.

Exercice 11

On propose le code suivant OCaml pour le tri fusion :

```

1 (* renvoie deux listes l1 l2 de même taille à 1 près,
2   contenant les éléments de l *)
3 let rec separer (l: 'a list) : 'a list * 'a list =
4   match l with
5   | [] -> [], []
6   | [x] -> [x], []
7   | x::y::q -> let l1, l2 = separer q in x::l1, y::l2
8
9 (* Fusionne l1 et l2 deux listes supposées triées par ordre croissant *)
10 let rec fusion (l1: 'a list) (l2: 'a list) : 'a list =
11   match l1, l2 with
12   | [], _ -> l2
13   | _, [] -> l1
14   | x1::q1, x2::q2 -> if x1 < x2 then x1 :: fusion q1 l2
15                       else x2 :: fusion l1 q2
16
17 let rec tri_fusion (l : 'a list) : 'a list =

```

```

18 | match l with
19 | [] | [x] -> l
20 | _ -> let l1, l2 = separer l in fusion (tri_fusion l1) (tri_fusion l2)

```

Q1. Montrer la correction des trois fonctions

Q2. Donner la complexité des fonctions `separer` et `fusion` en fonction de la taille de leurs arguments.

Q3. En notant C_n le nombre d'opérations de `tri_fusion` sur une liste de taille n , justifier que C_n vérifie la relation de récurrence suivante :

$$C_n = C_{\lfloor \frac{n}{2} \rfloor} + C_{\lceil \frac{n}{2} \rceil} + \Theta(n)$$

Q4. Étudier la relation suivante simplifiée :

$$C_n = 2C_{\frac{n}{2}} + n$$

en calculant d'abord C_{2^p} pour p entier. On admettra que $(C_n)_{n \in \mathbb{N}}$ est croissante. En conclure la complexité du tri fusion.

Q5. Retrouver cette complexité en étudiant l'arbre d'appel de `tri_fusion`.